Komputeralgebrai algoritmusok

Járai Antal

Ezek a programok csak szemléltetésre szolgálnak.

▶ 1. Történet

▶ 2. Algebrai alapok

▶ 3. Normál formák, reprezentáció

▶ 4. Aritmetika

▶ 5. Kínai maradékolás

▶ 6. Newton-iteráció, Hensel-felemelés

▶ 7. Legnagyobb közös osztó

▶ 8. Faktorizálás

▶ 9. Egyenletrendszerek

▶ 10. Gröbner-bázisok

▼ 11. Racionális törtfüggvények integrálása

```
> restart;
```

▼ E 11.1. Példa.

```
> diff(1/(x+1),x);
```

$$-\frac{1}{(x+1)^2} \tag{11.1.1}$$

E 11.2. Példa.

```
> int(1/(x^3+x),x);
```

$$\ln(x) - \frac{1}{2}\ln(x^2+1) \tag{11.2.1}$$

E 11.3. Példa.

```
> int(1/(x^2-2),x);
```

$$-\frac{1}{2}\sqrt{2}\,\mathrm{arctanh}\left(\frac{1}{2}x\sqrt{2}\right) \tag{11.3.1}$$

```
> diff(2^(1/2)/4*ln(x-2^(1/2))-2^(1/2)/4*ln(x+2^(1/2)),x);
  simplify(%);
```

$$\frac{1}{4}\frac{\sqrt{2}}{x-\sqrt{2}} - \frac{1}{4}\frac{\sqrt{2}}{x+\sqrt{2}}$$

$$\frac{1}{x^2-2} \tag{11.3.2}$$

A 11.1. Algoritmus.

```
> SquareFree:=proc(a,x) local i,out,b,c,y,z,w;
  i:=1; out:=[]; b:=diff(a,x);
  c:=gcd(a,b); w:=quo(a,c,x);
  while c<>1 do
    y:=gcd(w,c);
    z:=quo(w,y,x);
    out:=[op(out),z];
    i:=i+1;
    w:=y; c:=quo(c,y,x);
  od; out:=[op(out),w]; end;
```

SquareFree := **proc**(*a*, *x*) (11.4.1)

```
    local i, out, b, c, y, z, w;
    i := 1;
    out := [ ];
    b := diff(a, x);
    c := gcd(a, b);
    w := quo(a, c, x);
    while c <> 1 do
        y := gcd(w, c);
        z := quo(w, y, x);
        out := [op(out), z];
```

```
        i := i + 1;
        w := y;
        c := quo(c, y, x)
    end do;
    out := [op(out), w]
end proc
```

```
> SquareFree((-12*x^3+9*x+3)/(-12),x);
```

$$\left[x-1, \frac{1}{2}+x\right] \tag{11.4.2}$$

```
> sqrfree(-12*x^3+9*x+3);
```

$$[-3, [[x-1, 1], [1+2x, 2]]] \tag{11.4.3}$$

```
> PolynomialDiophant:=proc(a,b,r,x) local y,z,q;
  gcdex(a,b,x,'y','z'); q:=quo(y*r,b,x);
  [expand(y*r-q*b),expand(z*r+q*a)] end;
```

(11.4.4)

```
PolynomialDiophant := proc(a, b, r, x)
    local y, z, q;
    gcdex(a, b, x, 'y', 'z');
    q := quo(y*r, b, x);
    [expand(y*r - q*b), expand(z*r + q*a)]
end proc
```

```
> PartialFractions1:=proc(r,L,x) local a,b,l,i,c;
  l:=nops(L); if l<2 then return([[r,L[1]]]) fi;
  a:=1; for i to l-1 do a:=a*L[i]^i od; b:=L[l]^l;
  c:=PolynomialDiophant(a,b,r,x);
  [op(PartialFractions1(c[2],L[1..l-1],x)),[c[1],L[l]]];
  end;
```

(11.4.5)

```
PartialFractions1 := proc(r, L, x)
    local a, b, l, i, c;
    l := nops(L);
    if l < 2 then
        return [[r, L[1]]]
    end if;
    a := 1;
    for i to l - 1 do
        a := a*L[i]^i
    end do;
    b := L[l]^l;
    c := PolynomialDiophant(a, b, r, x);
    [op(PartialFractions1(c[2], L[1 .. l - 1], x)), [c[1], L[l]]]
end proc
```

```
> PartialFractions2:=proc(r,q,e,x) local a,b,l,i,u,v;
  if e<2 then return([r]) fi;
  u:=quo(r,q,x,v);
  [op(PartialFractions2(u,q,e-1,x)),v];
  end;
```

PartialFractions2 := **proc**(*r*, *q*, *e*, *x*) (11.4.6)

local *a*, *b*, *l*, *i*, *u*, *v*;

if $e < 2$ **then**

return [*r*]

end if;

u := *quo*(*r*, *q*, *x*, *v*);

[*op*(*PartialFractions2*(*u*, *q*, $e - 1$, *x*)), *v*]

end proc

```
> PartialFractions:=proc(r,L,x) local i,LL,LLL;
  LL:=PartialFractions1(r,L,x); LLL:=[];
  for i to nops(LL) do
    LLL:=[op(LLL),[LL[i][2],PartialFractions2(LL[i][1],LL[i][2]
  ,i,x)]];
  od; end;
>
```

PartialFractions := **proc**(*r*, *L*, *x*) (11.4.7)

local *i*, *LL*, *LLL*;

LL := *PartialFractions1*(*r*, *L*, *x*);

LLL := [ ];

for *i* **to** *nops*(*LL*) **do**

LLL := [*op*(*LLL*), [*LL*[*i*][2], *PartialFractions2*(*LL*[*i*][1], *LL*[*i*][2], *i*, *x*)]]

end do

end proc

```
> HermiteReduction:=proc(p,q,x) local pp,rp,r,qq,ip,i,qi,ri,n,
  c;
  pp:=quo(p,q,x); r:=rem(p,q,x);
  qq:=SquareFree(q,x);
  r:=PartialFractions(r,qq,x);
  rp:=0; ip:=0;
  for i to nops(r) do
    qi:=r[i][1]; ri:=r[i][2]; n:=i;
    while n>1 do
      if ri[n]<>0 then
        c:=PolynomialDiophant(qi,diff(qi,x),ri[n],x);
        rp:=rp-c[2]/(n-1)/qi^(n-1);
        ri[n-1]:=ri[n-1]+c[1]+diff(c[2],x)/(n-1);
      fi; n:=n-1;
```

```
    od;
    ip:=ip+ri[1]/qi;
  od; rp+int(pp,x)+Int(ip,x); end;
```

$HermiteReduction := \mathbf{proc}(p, q, x)$ (11.4.8)

$\quad \mathbf{local}\ pp, rp, r, qq, ip, i, qi, ri, n, c;$

$\quad pp := quo(p, q, x);$

$\quad r := rem(p, q, x);$

$\quad qq := SquareFree(q, x);$

$\quad r := PartialFractions(r, qq, x);$

$\quad rp := 0;$

$\quad ip := 0;$

$\quad \mathbf{for}\ i\ \mathbf{to}\ nops(r)\ \mathbf{do}$

$\qquad qi := r[i][1];$

$\qquad ri := r[i][2];$

$\qquad n := i;$

$\qquad \mathbf{while}\ 1 < n\ \mathbf{do}$

$\qquad\quad \mathbf{if}\ ri[n] <> 0\ \mathbf{then}$

$\qquad\qquad c := PolynomialDiophant(qi, diff(qi, x), ri[n], x);$

$\qquad\qquad rp := rp - c[2] / ((n-1) * qi \wedge (n-1));$

$\qquad\qquad ri[n-1] := ri[n-1] + c[1] + (diff(c[2], x)) / (n-1)$

$\qquad\quad \mathbf{end\ if};$

$\qquad\quad n := n - 1$

$\qquad \mathbf{end\ do};$

$\qquad ip := ip + ri[1] / qi$

$\quad \mathbf{end\ do};$

$\quad rp + int(pp, x) + Int(ip, x)$

$\mathbf{end\ proc}$

E 11.4. Példa.

```
> fp:=441*x^7+780*x^6-2861*x^5+4085*x^4+7695*x^3+3713*x^2
  -43253*x+24500;
```

$$fp := 441\,x^7 + 780\,x^6 - 2861\,x^5 + 4085\,x^4 + 7695\,x^3 + 3713\,x^2 - 43253\,x + 24500 \quad (11.5.1)$$

```
> fq:=9*x^6+6*x^5-65*x^4+20*x^3+135*x^2-154*x+49;
```

$$fq := 9\,x^6 + 6\,x^5 - 65\,x^4 + 20\,x^3 + 135\,x^2 - 154\,x + 49 \quad (11.5.2)$$

```
> fp:=fp/9; fq:=fq/9; fr:=rem(fp,fq,x); f:=fp/fq;
```

$$fp := 49\,x^7 + \frac{260}{3}\,x^6 - \frac{2861}{9}\,x^5 + \frac{4085}{9}\,x^4 + 855\,x^3 + \frac{3713}{9}\,x^2 - \frac{43253}{9}\,x$$

$$+\frac{24500}{9}$$

$$fq := x^6+\frac{2}{3}x^5-\frac{65}{9}x^4+\frac{20}{9}x^3+15x^2-\frac{154}{9}x+\frac{49}{9}$$

$$fr := \frac{21854}{9}+735x^4+441x^2-\frac{12446}{3}x$$

$$f := \frac{1}{x^6+\frac{2}{3}x^5-\frac{65}{9}x^4+\frac{20}{9}x^3+15x^2-\frac{154}{9}x+\frac{49}{9}}\left(49x^7+\frac{260}{3}x^6-\frac{2861}{9}x^5+\frac{4085}{9}x^4+855x^3+\frac{3713}{9}x^2-\frac{43253}{9}x+\frac{24500}{9}\right) \quad (11.5.3)$$

```
> qq:=SquareFree(fq,x);
```

$$qq := \left[1, x+\frac{7}{3}, 1, x-1\right] \quad (11.5.4)$$

```
> PartialFractions1(fr,qq,x);
```

$$\left[[0, 1], \left[294, x+\frac{7}{3}\right], [0, 1], \left[392+441x^2-882x, x-1\right]\right] \quad (11.5.5)$$

```
> PartialFractions2(392+441*x^2-882*x,x-1,4,x);
```

$$[0, 441, 0, -49] \quad (11.5.6)$$

```
> PartialFractions(fr,qq,x);
```

$$\left[[1, [0]], \left[x+\frac{7}{3}, [0, 294]\right], [1, [0, 0, 0]], [x-1, [0, 441, 0, -49]]\right] \quad (11.5.7)$$

```
> convert(f,parfrac,x,sqrfree);
```

$$49x+54+\frac{441}{(x-1)^2}-\frac{49}{(x-1)^4}+\frac{2646}{(3x+7)^2} \quad (11.5.8)$$

```
> PolynomialDiophant(x+7/3,1,294,x);
```

$$[0, 294] \quad (11.5.9)$$

```
> int(294/(x+7/3)^2,x);
```

$$-\frac{294}{x+\frac{7}{3}} \quad (11.5.10)$$

```
> int(441/(x-1)^2-49/(x-1)^4,x);
```

$$-\frac{441}{x-1}+\frac{49}{3(x-1)^3} \quad (11.5.11)$$

```
> HermiteReduction(fp,fq,x);
```

$$-\frac{294}{x+\frac{7}{3}}+\frac{49}{3(x-1)^3}-\frac{441}{x-1}+\frac{49}{2}x^2+54x+\int 0\,dx \quad (11.5.12)$$

E 11.5. Példa.

```
> gp:=36*x^6+126*x^5+183*x^4+13807/6*x^3-407*x^2-3242/5*
  x+3044/15;;
```

$$gp := 36\,x^6 + 126\,x^5 + 183\,x^4 + \frac{13807}{6}\,x^3 - 407\,x^2 - \frac{3242}{5}\,x + \frac{3044}{15} \quad (11.6.1)$$

```
> gq:=(x^2+7/6*x+1/3)^2*(x-2/5)^3;
```

$$gq := \left(x^2 + \frac{7}{6}\,x + \frac{1}{3}\right)^2 \left(x - \frac{2}{5}\right)^3 \quad (11.6.2)$$

```
> g:=gp/gq;
```

$$g := \frac{36\,x^6 + 126\,x^5 + 183\,x^4 + \frac{13807}{6}\,x^3 - 407\,x^2 - \frac{3242}{5}\,x + \frac{3044}{15}}{\left(x^2 + \frac{7}{6}\,x + \frac{1}{3}\right)^2 \left(x - \frac{2}{5}\right)^3} \quad (11.6.3)$$

```
> convert(g,parfrac,x,sqrfree);
```

$$\frac{1770}{(5\,x-2)^2} + \frac{4320}{(5\,x-2)^3} + \frac{187255}{16\,(5\,x-2)} + \frac{1}{16}\,\frac{-346625 - 221250\,x}{6\,x^2 + 7\,x + 2} + \frac{-47025 - 79650\,x}{(6\,x^2 + 7\,x + 2)^2} \quad (11.6.4)$$

A 11.2. Algoritmus.

```
> HorowitzReduction:=proc(p,q,x) local pop,pp,d,b,m,n,a,aa,c,
  cc,r,i,j,e,s;
  pop:=quo(p,q,x); pp:=rem(p,q,x);
  d:=gcd(q,diff(q,x)); b:=quo(q,d,x);
  m:=degree(b); n:=degree(d);
  aa:=sum(a[i]*x^i,i=0..m-1);
  cc:=sum(c[i]*x^i,i=0..n-1);
  r:=expand(b*diff(cc,x)-cc*quo(b*diff(d,x),d,x)+d*aa);
  for i from 0 to m+n-1 do e[i]:=coeff(pp,x,i)=coeff(r,x,i);
  od;
  s:=solve([e[j]$j=0..m+n-1],[a[j]$j=0..m-1,c[j]$j=0..n-1]);
  aa:=sum(a[j]*x^j,j=0..m-1); aa:=subs(op(s),aa);
  cc:=sum(c[j]*x^j,j=0..n-1); cc:=subs(op(s),cc);
  cc/d+Int(pop,x)+Int(aa/b,x);
  end;
```

HorowitzReduction := **proc**(*p*, *q*, *x*) (11.7.1)

local *pop*, *pp*, *d*, *b*, *m*, *n*, *a*, *aa*, *c*, *cc*, *r*, *i*, *j*, *e*, *s*;

pop := *quo*(*p*, *q*, *x*);

pp := *rem*(*p*, *q*, *x*);

d := *gcd*(*q*, *diff*(*q*, *x*));

```
    b := quo(q, d, x);
    m := degree(b);
    n := degree(d);
    aa := sum(a[i]*x^i, i = 0..m − 1);
    cc := sum(c[i]*x^i, i = 0..n − 1);
    r := expand(b*(diff(cc, x)) − cc*quo(b*(diff(d, x)), d, x)
    + d*aa);
    for i from 0 to m + n − 1 do
        e[i] := coeff(pp, x, i) = coeff(r, x, i)
    end do;
    s := solve([$(e[j], j = 0..m
    + n − 1)], [$(a[j], j = 0..m − 1), $(c[j], j = 0..n − 1)]);
    aa := sum(a[j]*x^j, j = 0..m − 1);
    aa := subs(op(s), aa);
    cc := sum(c[j]*x^j, j = 0..n − 1);
    cc := subs(op(s), cc);
    cc / d + Int(pop, x) + Int(aa / b, x)
end proc
```

E 11.6. Példa.

```
> debug(HorowitzReduction); HorowitzReduction(fp,fq,x);
```

HorowitzReduction

```
{--> enter HorowitzReduction, args = 49*x^7+260/3*x^6
-2861/9*x^5+4085/9*x^4+855*x^3+3713/9*x^2-43253/9*
x+24500/9, x^6+2/3*x^5-65/9*x^4+20/9*x^3+15*x^2-154/9*
x+49/9, x
```

$$pop := 49\,x + 54$$

$$pp := \frac{21854}{9} + 735\,x^4 + 441\,x^2 - \frac{12446}{3}\,x$$

$$d := -\frac{7}{3} + 6\,x - 4\,x^2 - \frac{2}{3}\,x^3 + x^4$$

$$b := x^2 + \frac{4}{3}\,x - \frac{7}{3}$$

$$m := 2$$

$$n := 4$$

$$aa := a_0 + a_1\,x$$

$$cc := c_0 + c_1\,x + c_2\,x^2 + c_3\,x^3$$

$$r := -\frac{10}{3} c_2 x^2 - \frac{7}{3} a_0 - \frac{7}{3} a_1 x - \frac{14}{3} c_1 x - 2 c_3 x^3 - \frac{14}{3} c_2 x - 7 c_3 x^2 - 3 x^2 c_1 - 2 x^3 c_2 - x^4 c_3 - 4 c_0 x + 6 x a_0 + 6 a_1 x^2 - 4 x^2 a_0 - 4 x^3 a_1 - \frac{2}{3} x^3 a_0 - \frac{2}{3} x^4 a_1 + x^4 a_0 + x^5 a_1 - 6 c_0 - \frac{7}{3} c_1$$

$$e_0 := \frac{21854}{9} = -\frac{7}{3} a_0 - 6 c_0 - \frac{7}{3} c_1$$

$$e_1 := -\frac{12446}{3} = -\frac{7}{3} a_1 - \frac{14}{3} c_1 - \frac{14}{3} c_2 - 4 c_0 + 6 a_0$$

$$e_2 := 441 = -\frac{10}{3} c_2 - 7 c_3 - 3 c_1 + 6 a_1 - 4 a_0$$

$$e_3 := 0 = -2 c_3 - 2 c_2 - 4 a_1 - \frac{2}{3} a_0$$

$$e_4 := 735 = -c_3 - \frac{2}{3} a_1 + a_0$$

$$e_5 := 0 = a_1$$

$$s := \left[\left[a_0 = 0,\ a_1 = 0,\ c_0 = -\frac{6272}{9},\ c_1 = \frac{2254}{3},\ c_2 = 735,\ c_3 = -735\right]\right]$$

$$aa := a_0 + a_1 x$$

$$aa := 0$$

$$cc := c_0 + c_1 x + c_2 x^2 + c_3 x^3$$

$$cc := -\frac{6272}{9} + \frac{2254}{3} x + 735 x^2 - 735 x^3$$

$$\frac{-\frac{6272}{9} + \frac{2254}{3} x + 735 x^2 - 735 x^3}{-\frac{7}{3} + 6 x - 4 x^2 - \frac{2}{3} x^3 + x^4} + \int (49 x + 54)\, dx + \int 0\, dx$$

```
<-- exit HorowitzReduction (now at top level) = (
-6272/9+2254/3*x+735*x^2-735*x^3)/(-7/3+6*x-4*x^2-2/3*x^3+
x^4)+Int(49*x+54, x)+Int(0, x)}
```

$$\frac{-\frac{6272}{9} + \frac{2254}{3} x + 735 x^2 - 735 x^3}{-\frac{7}{3} + 6 x - 4 x^2 - \frac{2}{3} x^3 + x^4} + \int (49 x + 54)\, dx + \int 0\, dx \qquad (11.8.1)$$

E 11.7. Példa.

```
> debug(HorowitzReduction); HorowitzReduction(gp,gq,x);
```

HorowitzReduction

```
{--> enter HorowitzReduction, args = 36*x^6+126*x^5+183*
x^4+13807/6*x^3-407*x^2-3242/5*x+3044/15, (x^2+7/6*x+1/3)
^2*(x-2/5)^3, x
```

$$pop := 0$$

$$pp := 36\,x^6 + 126\,x^5 + 183\,x^4 + \frac{13807}{6}\,x^3 - 407\,x^2 - \frac{3242}{5}\,x + \frac{3044}{15}$$

$$d := \left(x^2 + \frac{7}{6}\,x + \frac{1}{3}\right)\left(x - \frac{2}{5}\right)^2$$

$$b := x^3 + \frac{23}{30}\,x^2 - \frac{2}{15}\,x - \frac{2}{15}$$

$$m := 3$$

$$n := 4$$

$$aa := a_0 + a_1\,x + a_2\,x^2$$

$$cc := c_0 + c_1\,x + c_2\,x^2 + c_3\,x^3$$

$$r := -3\,x^3\,c_1 - 2\,x^4\,c_2 - x^5\,c_3 - \frac{29}{15}\,x^2\,c_1 - \frac{7}{6}\,x^3\,c_2 - \frac{2}{5}\,x^4\,c_3 - 4\,c_0\,x^2 - \frac{27}{10}\,c_0\,x + x^4\,a_0 + x^5\,a_1 + x^6\,a_2 + \frac{11}{30}\,x^3\,a_0 + \frac{11}{30}\,x^4\,a_1 + \frac{11}{30}\,x^5\,a_2 - \frac{11}{25}\,x^2\,a_0 - \frac{11}{25}\,x^3\,a_1 - \frac{11}{25}\,x^4\,a_2 - \frac{2}{25}\,x\,a_0 - \frac{2}{25}\,a_1\,x^2 - \frac{2}{25}\,a_2\,x^3 + \frac{4}{75}\,a_0 + \frac{4}{75}\,a_1\,x + \frac{4}{75}\,a_2\,x^2 - \frac{1}{3}\,c_1\,x - \frac{7}{15}\,c_2\,x^2 - \frac{3}{5}\,c_3\,x^3 - \frac{4}{15}\,c_2\,x - \frac{2}{5}\,c_3\,x^2 - \frac{1}{5}\,c_0 - \frac{2}{15}\,c_1$$

$$e_0 := \frac{3044}{15} = \frac{4}{75}\,a_0 - \frac{1}{5}\,c_0 - \frac{2}{15}\,c_1$$

$$e_1 := -\frac{3242}{5} = -\frac{27}{10}\,c_0 - \frac{2}{25}\,a_0 + \frac{4}{75}\,a_1 - \frac{1}{3}\,c_1 - \frac{4}{15}\,c_2$$

$$e_2 := -407 = -\frac{29}{15}\,c_1 - 4\,c_0 - \frac{11}{25}\,a_0 - \frac{2}{25}\,a_1 + \frac{4}{75}\,a_2 - \frac{7}{15}\,c_2 - \frac{2}{5}\,c_3$$

$$e_3 := \frac{13807}{6} = -3\,c_1 - \frac{7}{6}\,c_2 + \frac{11}{30}\,a_0 - \frac{11}{25}\,a_1 - \frac{2}{25}\,a_2 - \frac{3}{5}\,c_3$$

$$e_4 := 183 = -2\,c_2 - \frac{2}{5}\,c_3 + a_0 + \frac{11}{30}\,a_1 - \frac{11}{25}\,a_2$$

$$e_5 := 126 = -c_3 + a_1 + \frac{11}{30}\,a_2$$

$$e_6 := 36 = a_2$$

$$s := \left[\left[a_0 = \frac{3549}{2}, a_1 = 1167, a_2 = 36, c_0 = \frac{7142}{25}, c_1 = -\frac{31018}{25}, c_2 = \frac{39547}{50}, c_3 = \frac{5271}{5}\right]\right]$$

$$aa := a_0 + a_1\,x + a_2\,x^2$$

$$aa := \frac{3549}{2} + 1167\,x + 36\,x^2$$

$$cc := c_0 + c_1\,x + c_2\,x^2 + c_3\,x^3$$

$$cc := \frac{7142}{25} - \frac{31018}{25}\,x + \frac{39547}{50}\,x^2 + \frac{5271}{5}\,x^3$$

$$\frac{\frac{7142}{25} - \frac{31018}{25}\,x + \frac{39547}{50}\,x^2 + \frac{5271}{5}\,x^3}{\left(x^2 + \frac{7}{6}\,x + \frac{1}{3}\right)\left(x - \frac{2}{5}\right)^2} + \int 0\,\mathrm{d}x + \int \frac{\frac{3549}{2} + 1167\,x + 36\,x^2}{x^3 + \frac{23}{30}\,x^2 - \frac{2}{15}\,x - \frac{2}{15}}\,\mathrm{d}x$$

```
<-- exit HorowitzReduction (now at top level) = (7142/25
-31018/25*x+39547/50*x^2+5271/5*x^3)/((x^2+7/6*x+1/3)*(x
-2/5)^2)+Int(0, x)+Int((3549/2+1167*x+36*x^2)/(x^3+23/30*
x^2-2/15*x-2/15), x)}
```

$$\frac{\frac{7142}{25} - \frac{31018}{25}\,x + \frac{39547}{50}\,x^2 + \frac{5271}{5}\,x^3}{\left(x^2 + \frac{7}{6}\,x + \frac{1}{3}\right)\left(x - \frac{2}{5}\right)^2} + \int 0\,\mathrm{d}x + \int \frac{\frac{3549}{2} + 1167\,x + 36\,x^2}{x^3 + \frac{23}{30}\,x^2 - \frac{2}{15}\,x - \frac{2}{15}}\,\mathrm{d}x \qquad (11.9.1)$$

A 11.3. Algoritmus.

```
>
```

E 11.8. Példa.

```
> a:=1; b:=x^3+x; resultant(a-z*diff(b,x),b,x);
```

$$a := 1$$

$$b := x^3 + x$$

$$(1 - z)(2z + 1)^2 \quad (11.11.1)$$

```
> gcd(a-1*diff(b,x),b); gcd(a+1/2*diff(b,x),b);
```

$$x$$

$$x^2 + 1 \quad (11.11.2)$$

E 11.9. Példa.

```
> a:=1; b:=x^2-2; resultant(a-z*diff(b,x),b,x);
```

$$a := 1$$

$$b := x^2 - 2$$

$$-8z^2 + 1 \quad (11.12.1)$$

```
> alpha:=2^(1/2)/4; gcd(a+alpha*diff(b,x),b); gcd(a-alpha*diff
  (b,x),b);
```

$$\alpha := \frac{1}{4}\sqrt{2}$$

$$\sqrt{2} + x$$

$$-\sqrt{2} + x \quad (11.12.2)$$

E 11.10. Példa.

```
> a:=36*x^2+1167*x+3549/2; b:=x^3+23/30*x^2-2/15*x-2/15;
  resultant(a-z*diff(b,x),b,x);
```

$$a := 36x^2 + 1167x + \frac{3549}{2}$$

$$b := x^3 + \frac{23}{30}x^2 - \frac{2}{15}x - \frac{2}{15}$$

$$\frac{16}{625}z^3 - \frac{576}{625}z^2 - \frac{20872009}{16}z + 2730177900 \quad (11.13.1)$$

```
> factor(%);
```

$$\frac{1}{10000}(16z - 37451)(z + 8000)(16z - 91125) \quad (11.13.2)$$

```
> gcd(a+8000*diff(b,x),b); gcd(a-91125/16*diff(b,x),b);
  gcd(a-37451/16*diff(b,x),b);
```

$$\frac{1}{2} + x$$

$$\frac{2}{3} + x$$

(11.13.3)

$$-\frac{2}{5}+x \tag{11.13.3}$$

E 11.11. Példa.

```
> a:=7*x^13+10*x^8+4*x^7-7*x^6-4*x^3-4*x^2+3*x+3;
  b:=x^14-2*x^8-2*x^7-2*x^4-4*x^3-x^2+2*x+1;
  resultant(a-z*diff(b,x),b,x);
```

$$a := 7\,x^{13}+10\,x^8+4\,x^7-7\,x^6-4\,x^3-4\,x^2+3\,x+3$$

$$b := x^{14}-2\,x^8-2\,x^7-2\,x^4-4\,x^3-x^2+2\,x+1$$

$$145107402137728+4063007259856384\,z+44693079858420224\,z^2+227528406551957504\,z^3+373796667906787328\,z^4-110513797468093648\,z^5-396549508561983078 4\,z^6+341756953514776985 6\,z^7+1586198034247932313 6\,z^8-1768220759489498316 8\,z^9-2392298674603438899 2\,z^{10}+5824727207730112102 4\,z^{11}-4576571377502230937 6\,z^{12}+1664207773637174886 4\,z^{13}-2377439676624535552\,z^{14} \tag{11.14.1}$$

```
> factor(%);
```

$$-145107402137728\left(4\,z^2-4\,z-1\right)^7 \tag{11.14.2}$$

```
> alpha:=(1+2^(1/2))/2;
  gcd(a-alpha*diff(b,x),b); gcd(a-(1-alpha)*diff(b,x),b);
```

$$\alpha := \frac{1}{2}+\frac{1}{2}\sqrt{2}$$

$$x^7-\sqrt{2}\,x^2-x-\sqrt{2}\,x-1$$

$$x^7+\sqrt{2}\,x^2-x+\sqrt{2}\,x-1 \tag{11.14.3}$$

A 11.4. Algoritmus.

```
>
```

E 11.12. Példa.

```
> a:=6*x^5+6*x^4-8*x^3-18*x^2+8*x+8;
  b:=x^6-5*x^4-8*x^3-2*x^2+2*x+1;
  resultant(a-z*diff(b,x),b,x);
```

$$a := 6\,x^5+6\,x^4-8\,x^3-18\,x^2+8\,x+8$$

$$b := x^6-5\,x^4-8\,x^3-2\,x^2+2\,x+1$$

$$-1453248\,z^6+8719488\,z^5-8719488\,z^4-23251968\,z^3+17438976\,z^2 \tag{11.16.1}$$

$$+34877952\,z+11625984$$

```
> factor(%);
```

$$-1453248\left(z^2-2\,z-2\right)^3 \qquad (11.16.2)$$

E 11.13. Példa.

```
> a:=2*x^5-19*x^4+60*x^3-159*x^2+50*x+11;
  b:=x^6-13*x^5+58*x^4-85*x^3-66*x^2-17*x+1;
  resultant(a-z*diff(b,x),b,x);
```

$$a := 2\,x^5-19\,x^4+60\,x^3-159\,x^2+50\,x+11$$

$$b := x^6-13\,x^5+58\,x^4-85\,x^3-66\,x^2-17\,x+1$$

$$-190107645728000\,z^6+380215291456000\,z^5-570322937184000\,z^4+190107645728000\,z^2-380215291456000\,z-190107645728000 \qquad (11.17.1)$$

```
> factor(%);
```

$$-190107645728000\left(z^3-z^2+z+1\right)^2 \qquad (11.17.2)$$

12. A Risch-algoritmus.